
Abstract

A high-level abstract-datatype-based constraint
modelling language opens the door to an automat-
able empirical determination — by a compiler —
of how to ‘best’ represent the variables of a com-
binatorial optimisation problem, based on (real-
life) training instances of the problem. In the ex-
treme case where no such training instances are
provided, such a compiler would simply be non-
deterministic. A first-order relational calculus with
sets is a good candidate for such a language, as
it gives rise to very natural and easy-to-maintain
models of combinatorial optimisation problems.

1 Introduction
In recent years, modelling languages based on some logic
with sets and relations have gained popularity in formal meth-
ods, witness theB [1] andZ [13] specification languages, the
ALLOY [9] object modelling language, and the Object Con-
straint Language (OCL) of UML . In database modelling, this
had been long advocated, most notably via entity-relation-
attribute (ERA) diagrams. This study examines whether con-
straint modelling can benefit from the same ideas.

Sets and set expressions recently started appearing as mod-
elling devices in some constraint programming languages,
with set variables often implemented by the set interval rep-
resentation[7]. In the absence of such an explicit set concept,
modellers usually represent a set variable as an array of 0/1
integer variables, indexed by the domain of the set. In terms
of propagation, the set interval representation is equivalent to
the 0/1 representation. The latter of course consumes more
memory, but is able to support more set expressions and con-
straints. Both options are restricted to finite sets.

Relations have not received much attention yet in con-
straint programming languages, except the particular caseof
a total function, via the concept of array. Indeed, a total func-
tion f can be represented either as a 1D array of variables
over the range off , indexed by its domain, or as a 2D array of
0/1 variables, indexed by the domainand range off , or even
with some redundancy. Alternatively, we have recently advo-
cated that total functions should be supported by an abstract
datatype, so that it is the compiler that must choose a suitable

representation, depending on the context[4] and given (real-
life) training instances[8]. Other than retrieving the (unique)
image under a total function of a domain element, there has
(thus) been no support of relational expressions.

I here make two claims. First, a high-level abstract-
datatype-based constraint modelling language opens the door
to an automatable empirical determination — by a compiler
— of how to ‘best’ represent the variables of a combinatorial
optimisation problem, based on (real-life) training instances
of the problem. In the extreme case where no such train-
ing instances are provided, such a compiler would simply be
non-deterministic. Second, our previous work on an abstract
datatype for total functions[4; 8] can be usefully generalised
to supportany kind of relations, and a suitable first-order re-
lational calculus with sets is a good candidate for such a lan-
guage, as it gives rise to very natural and easy-to-maintain
models of combinatorial optimisation problems.

I here ignore the issue of how to parameterise a solver, say
by providing a suitable labelling heuristic, towards the solv-
ing of the modelled problem. For non-expert or lazy mod-
ellers, this task can also be left to compilers[5; 11]. I thus
here only aim at techniques that find the ‘best’ model for a
given solver, under itsdefault settings. A joint consideration
of the modelling and the solver parameterisation is omitted
here so as not to clutter the reported ideas.

As befits a workshop, this is partially a position paper, in
the sense that the denotational and operational formal seman-
tics of the proposed language have not been fully worked out
yet (but see[3]), and that a prototype of the advocated com-
piler is not available yet. My aims here only are to present the
notation, illustrate its elegance and the flexibility of itsmodels
by some examples, and argue that the advocated compilation
philosophy is feasible and useful.

This paper is organised as follows. In Section 2, I present
my relational notation for modelling combinatorial optimi-
sation problems. Next, I illustrate my claims on two such
problems, namely the Warehouse Location problem in Sec-
tion 3 and the Stable Marriage problem in Section 4. Finally,
in Section 5, I discuss related and future work.

2 Relational Modelling with ESRA

After discussing, in Section 2.1, the design decisions be-
hind the newESRA modelling language, I introduce, in Sec-
tion 2.2, its concepts, syntax, and semantics.



2.1 Design Decisions
In constraint satisfaction, much more effort has been directed
at efficiently solving the constraints than at facilitatingtheir
modelling. Constraint programming languages reflect this,as
their control structures and variable representation options are
usually quite low-level. This has significantly changed with
the advent of the Optimisation Programming Language (OPL)
[14], which provides a (nicely sugared) bigger subset of first-
order logic than the usual Horn clauses, plus enumerated sets
for dispensing with the frequent need of encoding everything
as integers, and direct-access arrays (of any dimension) in-
stead of the usual sequential-access lists. ButOPL does not go
far enough as a modelling language, in our opinion: set vari-
ables, relations, and some useful quantifiers are still missing,
and thus need to be represented at a rather low level, in unnat-
ural but standard ways (see below for examples). The solver
of OPL is excellent, especially due to its being a front-end
to both constraint and linear programming solvers, though a
more open parameterisation would be useful.

The key design decisions for our new constraint modelling
language — calledESRA — are as follows. We want to cap-
ture common modelling idioms in new abstract datatypes and
quantifiers, especially for sets and relations, so as to design a
truly high-level language. Computational completeness isnot
aimed at, as long as the notation is useful for elegantly mod-
elling a large number of combinatorial optimisation prob-
lems. LikeOPL, we (currently) do not support procedures,
and hence no procedure calls and no recursion. Similarly, like
OPL, we focus on finite domains, and support only bounded
quantification. In order to maximally sugar the first-order-
logic nature of the language, we adopt anOPL/ALLOY -like
‘lower-128 ASCII’ syntax, unlike the LATEX-requiring syntax
of Z, as well as anOPL-like JAVA-style declaration of the uni-
versally quantified input/output parameters.

Considering the excellent starting point thatOPL thus pro-
vides, it became natural to designESRA by extending a
streamlined (significant) subset ofOPL.1 The semantics2

of ESRA will be given in an implementation-independent
way, in two layers. Indeed, some features ofESRA are just
syntactic sugar for combinations of (a few) kernel features,
hence we will provide an operational semantics (by rewrite
rules) for the non-kernel features, and a set-oriented deno-
tational semantics for the kernel features. As shown in[3;
4], it is actually possible to give an operational semantics by
ESRA-to-OPL rewrite rules for theentire language.

2.2 Concepts, Syntax, and Semantics of ESRA

For reasons of space, I here only introduce the concepts of
ESRA that are actually illustrated in this paper. Also, I can
“only” give an informal semantics. The reader may monitor
[3] for a complete description of the full language.

Modelling the Instance Data and Variables. A primitive
type is either a finite enumeration of new constant identifiers,

1OPL leaves numerous opportunities for reducing its syntax
while increasing its power. For time reasons, we donot support
floats, scheduling, and some other features ofOPL.

2OPL doesnot (yet) have a published semantics.

or a finite range of integers, indicated by its lower and upper
bounds. Constant and variable identifiers can be any mix of
lowercase and uppercase letters. The only predefined primi-
tive types are the rangesnat andint , which are0:maxint
and -maxint:maxint , respectively, withmaxint being
the maximum representable integer.

An enumeration is viewed as aset, and can thus have sub-
sets. For this, the binaryf�g type constructor allows the con-
struction of thepowerset of a set as a new type, so that sets
can be declared of that type. Consider the powerset typefS
m:ng. ThenS must be a primitive type of the enumeration
kind, while the rangem:n is a multiplicity, stating that any
member of the powerset ofS must have betweenmandn el-
ements, wheremandn are natural-number expressions.

Relations are declared using the quaternary# relation type-
constructor. Consider the relation typeA m:n # p:q B .
ThenA andB must be primitive types, designating the two
participants of any relation of this type, withA being called
thedomain andB therange of such a relation, by extension of
the terminology for functions, which are just particular cases
of relations. The second and third arguments of# are mul-
tiplicities, with the following semantics: for every element
of A, there are betweenmandn elements ofB, and for ev-
ery element ofB, there are betweenp andq elements ofA
in such a relation.3 For partial and total functions,m:n is
0:1 and1:1 , respectively. For injections, surjections, and
bijections,p:q is 0:1 , 1:maxint , and1:1 , respectively.
Rather than elevating functions and their particular casesto
first-class concepts with specific syntax inESRA, I prefer
keeping the notation lean and leave their specialised handling
to the compiler. This has the further advantage that only the
multiplicities need to be changed during model maintenance
when a function becomes a relation (as seen in Section 4.3).

Like in OPL, (arrays of) instance data variables are declared
in a JAVA-style strongly typed syntax. Unlike inOPL, all in-
stance data are here read in at run-time from a data file.4 Prob-
lem variable declarations follow the same syntax, but are pre-
ceded by thevar keyword. The usage of arrays of problem
variables, though possible, is sometimes discouraged, as they
sometimes amount to a premature commitment to a low-level
representation of what essentially are relation variables. Due
to the (current) restrictions on relations, arrays arenot a re-
dundant feature ofESRA. All declarations denote universally
quantified variables, with the instance data ones expected to
be ground at solving-time and the problem ones expected to
still be variables then. Consider the following declarations:

nat MaxWives
enum Women, Men
{Men 1:maxint} MarriedMen

3I thus (currently) restrict the focus tobinary relations, be-
tween primitive types only. My convention is the opposite ofthe
UML /ALLOY one, say, where multiplicities are written in the other
order, with thesame semantics. That convention can howevernot be
usefully upgraded to relations of arity higher than 2, and I wanted to
leave the possibility for that extension open.

4I thus dispensed with all otherOPL forms of initialisation, and
thus with its ‘... ’ notation for data-file initialisation. I also dis-
pensed with its semicolon ‘; ’ after declarations.



int RankW[Women,Men]
var Women f
var Women 0:1 # 0:MaxWives Men Marriage

Here,MaxWives is declared to be a natural number, while
WomenandMenare enumerated sets. The setMarriedMen
is an element of the powerset ofMen, that is a subset ofMen,
and must be non-empty. The 2D integer arrayRankWis in-
dexed by the setsWomenandMen. The problem variablef
designates an element of the setWomen. Finally, the prob-
lem variableMarriage designates a relation over the sets
WomenandMen, such that there is at most one husband for
each woman, and at mostMaxWives wives for each man.

Modelling the Cost Function and the Constraints. Ex-
pressions are constructed in the usual way. Fornumeric
expressions, arguments are either integers, or the constant
maxint , or variables of typenat or int , and the usual
unary, binary, and aggregate arithmetic operators are avail-
able, such ascard for the cardinality of a set expression,
ord for the position of an identifier in an enumeration, the
infix ‘ +’ and ‘* ’ for the addition and multiplication of two
numeric expressions, andsum for the sum of a bounded (and
possibly filtered) number of numeric expressions.

For set expressions, arguments are either enumerated sets
or set variables, and the usual binary and aggregate set oper-
ators are available, such as the infixunion for the union of
two set expressions, and navigation expressions, explained
next. Let R be a relation of typeA m:n # p:q B . For
any element (or subset)a of A, the navigation expression
a.R designates the relational image ofa, that is the possi-
bly empty set of all elements inB that are related byR to (any
element in)a. If m:n is 1:1 , thena.R simply designates the
(unique) element ofB that is related to elementa of A. The
relation expression �Rdesignates thetranspose relation ofR,
which is thus of typeB p:q # m:n A . Transitive closure
will be added if suitable examples justify it. The elements of
a relation are represented asa#b pairs.

First-order logic formulas are also constructed in the usual
way, with some restrictions. Atoms are built from expres-
sions with the usual predicates, such as the infixin for set or
relation membership and the infix ‘<=’ for the ‘�’ inequality
between numeric expressions. Formulas are built from atoms
with the usual connectives and quantifiers, such asnot for
negation, the infix ‘&’ and ‘=>’ for conjunction and implica-
tion, andforall andexists for bounded (and possibly
filtered) universal and existential quantification. The usual
typing and precedence rules for operators and predicates ap-
ply. All binary operators associate to the left.

The cost function is a numeric expression that has to be
either minimised or maximised. Theconstraints on the prob-
lem variables are a conjunction of formulas. Consider the
following three constraints, given the declarations above:

card(Women.Marriage)
< card(Men.˜Marriage)

forall(m in Men)
card(m.˜Marriage) < MaxWives

exists(h in Men: f#h in Marriage)
card(h.˜Marriage) = 1

int MaintCost = ...;
enum WareHs ...;
enum Stores ...;
int Capacity[WareHs] = ...;
int SupplyCost[Stores,WareHs] = ...;
var WareHs Supply[Stores]; // C1
var int OpenWareHs[WareHs] in 0..1;
minimize

sum(s in Stores) SupplyCost[s,Supply[s]]
+ sum(w in WareHs) OpenWareHs[w]

* MaintCost
subject to {

forall(s in Stores) // link
OpenWareHs[Supply[s]]=1;

forall(w in WareHs) // C2
(sum(s in Stores) (Supply[s]=w))

<= Capacity[w]
};

Figure 1: PublishedOPL model of the Warehouse problem

In the first constraint, the navigation expression
Women.Marriage designates the set of married men,
while Men.�Marriage designates the set of married
women, which must thus be larger than the previous set. The
second constraint requires that no man be actually married
to the maximum legal number of wives. The third constraint
says that womanf must be married, to a manh who is
monogamous. Note the finite bounds ‘in Men ’ and the
optional filters (with problem variables), such as ‘f#h in
Marriage ’, in these quantifications.

3 The Warehouse Location Problem
In the Warehouse Location problem, a company considers
opening warehouses on some candidate locations in order to
supply its existing stores. Each candidate warehouse has the
same maintenance cost, and the supply cost to a store depends
on the warehouse. Each store must be supplied by exactly
one open warehouse (C1). Each candidate warehouse has a
capacity designating the maximum number of stores that it
can supply (C2). The objective is to determine which ware-
houses to open, and which of these warehouses should supply
the various stores, such that the sum of the maintenance and
supply costs is minimised. In more mathematical terms, the
sought supply relationship is atotal function (or: mapping)
from the set of stores into the set of warehouses, and the set
of warehouses to be opened is therange of that function.

A First OPL Model. This problem was first modelled
as a constraint program in the reference manual ofILOG
SOLVER 4.0 (in 1997), and then modelled inOPL in [14]: see
Figure 1 for (a syntactic and re-typeset variant of) that model.
No prior knowledge ofOPL is assumed here, as I explain all
its features used here. The maintenance cost is an integer read
in at run-time from a data file (which is indicated by the ‘... ’
notation), and so are the enumerated sets of candidate ware-
house locations and existing stores, the 1D array with the in-
teger capacities of the warehouses, and the 2D array with the



int MaintCost = ...;
enum WareHs ...;
enum Stores ...;
int Capacity[WareHs] = ...;
int SupplyCost[Stores,WareHs] = ...;
var WareHs Supply[Stores]; // C1
minimize

sum(s in Stores) SupplyCost[s,Supply[s]]
+ sum(w in WareHs)

(sum(s in Stores) (Supply[s]=w)>0)
* MaintCost

subject to {
forall(w in WareHs) // C2

(sum(s in Stores) (Supply[s]=w))
<= Capacity[w]

};

Figure 2: A secondOPL model of the Warehouse problem

integer supply costs to the stores from the warehouses. The
sought total function is modelled by a 1D array of variables
representing the (unique) warehouse that supplies each store,
so this choice fully captures constraintC1. The set of ware-
houses to be opened is modelled in a redundant way (because
it would suffice to retrieve the range of that function), namely
as a 1D array of 0/1 variables, such thatOpenWareHs[w] is
1 if and only if warehousew is opened.5 The objective state-
ment expresses that the addition of the sum of the actually
incurred supply costs and the sum of the maintenance costs
for the actually opened warehouses must be minimal. The
second sum is awkward, as the Booleans ofOpenWareHs
are re-interpreted as numeric weights. The first constraintis
a linking (or: channelling) constraint made necessary by the
redundant modelling. It expresses that a warehouse that is
actually supplying some store must be opened. The second
constraint capturesC2 using thehigher-order constraint fea-
ture of OPL, whereby a constraint wrapped in parentheses,
such as(Supply[s]=w) , is associated with a 0/1 integer
variable that becomes1 when the constraint can be proven
true, and0 when the constraint can be proven false, but that
remains undetermined otherwise. This may be a very power-
ful feature ofOPL, but reflects low-level thinking and should
be hidden from the modeller via a more high-level syntax.

A Second OPL Model. Let us design anotherOPL model,
which non-redundantly models the supply function, namely
just by the 1D array of variables representing the warehouse
that supplies each store. Figure 2 has the resulting model. The
linking constraint disappeared and the second sum of the cost
function became much more awkward, asOpenWareHs[w]
had to be replaced by a higher-order expression returning1
iff there exists a store supplied by warehousew. On the in-
stance data I tried, this model is an order of magnitude more
efficient (by all measures) than the one above. This proves
that redundancy elimination may pay off, but it may just as
well be the converse, as shown in Section 4.1. But this is
hard to guess, as human intuition may be weak here.

5OPL doesnot allow set variables, unlikeILOG SOLVER.

nat MaintCost
enum WareHs, Stores
nat Capacity[WareHs],

SupplyCost[Stores,WareHs]
var Stores 1:1 # nat WareHs Supply // C1
minimise

sum(s#w in Supply) SupplyCost[s,w]
+ card(Stores.Supply) * MaintCost

subject to {
forall(w in WareHs) // C2

card(w.˜Supply) <= Capacity[w]
}

Figure 3: AnESRA model of the Warehouse problem

An ESRA Model. Figure 3 shows anESRA model of the
problem. It is very different from the one we previously
gave[4], as it uses navigation expressions and has no redun-
dancy. The modelling of the instance data is similar to the
OPL models above. The sought supply relationship is mod-
elled as a relation and constrained to be a total function from
the stores into the warehouses, thereby capturing constraintC1. The elegance of the cost function reflects the freedom
from representation choices, with the navigation expression
Stores.Supply retrieving the set of warehouses that are
to be opened. The only constraint gracefully capturesC2,
using the navigation expressionw. �Supply to retrieve the
set of stores that warehousew supplies. This model can be
translated into variousOPL models, similar to the ones above.

4 The Stable Marriage Problem

I now consider the Stable Marriage problem, first in its origi-
nal version (in Section 4.1) and then in two modified versions
(in Sections 4.2 and 4.3). For the original version, I discuss
threeOPL models and exhibit a singleESRA model, which
could be mechanically compiled intoOPL models not unlike
the ones given. For the modified versions, I only exhibitESRA
models, showcasing the flexibility of the proposed relational
language, and leaving model maintenance at a lower level
(such asOPL) as an exercise to the reader.

4.1 Original Version

The original version of theStable Marriage problem can be
stated as follows. Consider a dating agency where an equal
numbern of women and men have signed up and are willing
(for whatever reason) to marry any opposite-sex person of
the group. They have ranked all possible spouses by decreas-
ing preference. Figure 4 has sample instance data, where a
lower rank means a higher preference. For instance, Hal is
Nat’s first choice, but it is Eve who is Hal’s first choice. The
objective is to match up the women and men such that all
marriages are stable. A marriage is said to bestable if, when-
ever spouses prefers some other partner, this partner prefers
her/his spouse tos. This means thats may be unhappy, but
s/he is bound to stay with her/his spouse. In more mathemat-
ical terms, the sought marriages form abijection between the
sets of women and men.



Hal Jim Bob Ian Guy
Nat 1 2 4 3 5
Eve 3 5 1 2 4
Pat 5 4 2 1 3
Sue 1 3 5 4 2
Val 4 2 3 5 1

Nat Eve Pat Sue Val
Hal 5 1 2 4 3
Jim 4 1 3 2 5
Bob 5 3 2 4 1
Ian 1 5 4 3 2
Guy 4 3 2 1 5

Figure 4: Rankings of the women for the men (top), and rank-
ings of the men for the women (bottom)

enum Women ...;
enum Men ...;
int RankW[Women,Men] = ...;
int RankM[Men,Women] = ...;
var Men Husband[Women];// all women marry
var Women Wife[Men]; // all men marry
solve {

forall(w in Women) // link 1: bijection
Wife[Husband[w]] = w;

forall(m in Men) // link 2: bijection
Husband[Wife[m]] = m;

// stability 1:
forall(w in Women, o in Men)

RankW[w,o] < RankW[w,Husband[w]]
=> RankM[o,Wife[o]] < RankM[o,w];

// stability 2:
forall(m in Men, o in Women)

RankM[m,o] < RankM[m,Wife[m]]
=> RankW[o,Husband[o]] < RankW[o,m]

};

Figure 5: PublishedOPL model of the original Stable Mar-
riage problem

A First OPL Model. This problem was first modelled
as a constraint program in the reference manual ofILOG
SOLVER 4.0 (in 1997), and then modelled inOPL in a sig-
nificantly simpler way in[14]: see Figure 5 for (a syntactic
and re-typeset variant of) that model. The enumerated sets of
women and men are to be read in at run-time, just like the
two 2D arrays of rankings of the women and men for each
other. Assertions could be added to ensure that both sets are
of sizen, and that each person’s ranking is a permutation of
the range 1:n. The sought marriages are modelled in a re-
dundant way, via two 1D arrays of variables representing the
(unique) husband of each woman and the (unique) wife of
each man, respectively. A linking constraint is thus neces-
sary to ensure that both total functions are the inverse of each
other, that is to achieve a bijection. To achieve better propa-
gation, this linking constraint is here expressed forboth func-
tions, requiring every person to be identical to the spouse of
their spouse. Finally, two stability constraints capture the def-
inition of stability of a marriage. This model showcases a key

enum Women ...;
enum Men ...;
int RankW[Women,Men] = ...;
int RankM[Men,Women] = ...;
var Women Wife[Men];
solve {

alldifferent(Wife); // bijection
// stability 1:
forall(w in Women, m,o in Men)

Wife[m]=w & RankW[w,o] < RankW[w,m]
=> RankM[o,Wife[o]] < RankM[o,w];

// stability 2:
forall(m,p in Men, o in Women)

Wife[p]=o & RankM[m,o]<RankM[m,Wife[m]]
=> RankW[o,p] < RankW[o,m]

};

Figure 6: A secondOPL model of the original Stable Marriage
problem

feature ofOPL, namely that expressions with variables, such
asWife[m] , can be used to index arrays (of variables). This
feature is crucial for concise and clear modelling of many
problems. Furthermore, the constraint-solving algorithmof
OPL uses such expressions to reduce the domains of the vari-
ables in the arrays via bi-directional propagation.

A Second OPL Model. Let us design anotherOPL model,
which less well exploits the features of theOPL solver. It
non-redundantly models the marriages, namely by a single
total function, that is a 1D array of variables representingthe
(unique) wife of each man, say. (Due to the symmetry in
the definition of marriage stability, the actual choice proba-
bly does not matter. However, in an optimisation version of
the problem, such as below, the choicewould matter if there
were some women or men that are more preferred than the
others.) Figure 6 has the resulting model. There is no need
for any linking constraints. To enforce the bijectiveness of
the function, it suffices to constrain all variables to be differ-
ent. Finally, the two stability constraints had to be rephrased
slightly, as there is now no direct way of retrieving the hus-
band of a given woman: this is achieved by iterating over all
the men. This model is probably less efficient, and this has
been the case with the instance data I tried.

A Third OPL Model. A third OPL model represents the
marriages in a 2D arrayMarriage of 0/1 integer variables,
indexed by the women and men, so thatMarriage[w,m]
is 1 iff womanw is married to manm. Figure 7 shows the re-
sulting model. Two bijectiveness constraints are necessary to
enforce that every person has exactly one spouse, that is that
there is exactly one1 in each row and in each column. The
two stability constraints are obtained from those of the sec-
ond model by the same technique as the stability constraints
of the second model were obtained from those of the first one.
This model is probably less efficient than the second one, and
this has been the case with the instance data I tried.



enum Women ...;
enum Men ...;
int RankW[Women,Men] = ...;
int RankM[Men,Women] = ...;
var int Marriage[Women,Men] in 0..1;
solve {

forall(w in Women) // bijection 1
sum(m in Men) Marriage[w,m] = 1;

forall(m in Men) // bijection 2
sum(w in Women) Marriage[w,m] = 1;

forall(w,p in Women, m,o in Men) {
// stability 1:
Marriage[w,m]=1 & Marriage[p,o]=1

& RankW[w,o] < RankW[w,m]
=> RankM[o,p] < RankM[o,w];

// stability 2:
Marriage[w,m]=1 & Marriage[p,o]=1

& RankM[m,p] < RankM[m,w]
=> RankW[p,o] < RankW[p,m];

}
};

Figure 7: A thirdOPL model of the original Stable Marriage
problem

enum Women, Men
nat RankW[Women,Men], RankM[Men,Women]
var Women 1:1 # 1:1 Men Marriage // bij.
solve {

forall(w#m, p#o in Marriage) {
RankW[w,o] < RankW[w,m] // stability 1

=> RankM[o,p] < RankM[o,w]
&
RankM[m,p] < RankM[m,w] // stability 2

=> RankW[p,o] < RankW[p,m]
}

}

Figure 8: AnESRA model of the original Stable Marriage
problem

An ESRA Model. Figure 8 shows anESRA model of the
problem. The modelling of the instance data is similar to the
OPL models above. The sought marriages are modelled as
a relation over the sets of women and men, such that it is a
bijection: for every woman, there is exactly one man, and for
every man, there is exactly one woman. There is no need for
any linking constraints. The two stability constraints directly
capture the definition of stability of a marriage, by iterating
over all pairs of married couples.

This model can be translated into variousOPL models, not
unlike the ones above, using the (possibly redundant) ways
of representing relations, and exploiting insights gainedfrom
thorough studies of bijections[12; 15]. The advantages of
abstract relational modelling are that the modeller need not
worry about the actual (and possibly redundant) representa-
tion of the relation and its multiplicity constraints, and that
the model is (thus) more flexible when the problem changes,
as shown next, as well as in Sections 4.2 and 4.3.

Model Maintenance. Paraphrasing the two model flexibil-
ity challenges raised in Section 8.4 of[10], where a different
bijection problem was examined, let us now consider the fol-
lowing two additional constraints for the problem.

First, suppose that (for whatever reason) we cannot have
both hSue, Jimi and hPat, Bobi as married couples. In the
first and secondOPL models, this can easily be expressed as
the following new constraint:

not(Wife[Jim]=Sue & Wife[Bob]=Pat)

or, probably more efficiently but at a lower level, as:

(Wife[Jim]=Sue) + (Wife[Bob]=Pat) <= 1

whereas in the thirdOPL model, this can also easily be ex-
pressed, say as:

not(Marriage[Sue,Jim]=1
& Marriage[Pat,Bob]=1)

or, probably more efficiently but less naturally, as:

Marriage[Sue,Jim]+Marriage[Pat,Bob] <= 1

In the ESRA model, the constraint can also easily be ex-
pressed, say as follows:

not(Sue#Jim in Marriage
& Pat#Bob in Marriage)

In other words, bothOPL andESRA easily handle this chal-
lenge. However, in a lower-level language thanOPL, the
equivalents of the first and secondOPL models make this new
constraint rather hard to express (see[10] for a solution).

Second, suppose that (for whatever reason) Bob must
marry a woman coming after the wife of Jim in the enumera-
tion of women. In the first and secondOPL models, this can
easily be expressed as the following new constraint:

ord(Wife[Bob]) > ord(Wife[Jim])

but in the thirdOPL model, this can be expressed only with
difficulty, say as:

sum(w in Women) Marriage[w,Bob] * ord(w)
> sum(w in Women) Marriage[w,Jim] *ord(w)

In the ESRA model, the constraint can easily be expressed,
say as follows:

ord(Bob.˜Marriage) > ord(Jim.˜Marriage)

In other words, this timesome OPL models arenot easy to
maintain, so onlyESRA easily handles the challenge, be-
cause its model is unencumbered with representation details.
Worse, in a lower-level language thanOPL, the equivalent
of the thirdOPL model makes this additional constraint even
harder to express (see[10] for a solution).

Model maintenance at the highESRA level thus reduces to
adapting to the new problem, with all representation (and thus
solving) issues left to the compiler. At lower levels, even with
OPL, model maintenance may become quite tedious, because
the already made representation choices have to be taken into
account and because the lower-level notation is more awk-
ward. Worse, a representation change, a redundancy elimi-
nation, or a redundancy introduction (such as a model inte-
gration or the addition of implied constraints) may “have to”
be operated, because it is unlikely that, for the considered



training instances or even in general, the ‘best’ representa-
tion is the same for bijections as for full relations, say. Such
model maintenance may become necessary not only when the
actual problem changes, but also because the distribution of
instances on which the model is deployed becomes different
from the training distribution used when the model was for-
mulated. But the modeller may be unwilling or unable to do
this experimentation for finding the ‘best’ model, or s/he may
be unaware of insights gained from a general empirical study,
such as on how to ‘best’ model bijections[12].

Fortunately, relations and their particular cases (such as
partial functions, total functions, injections, surjections, bi-
jections, and so on) are asingle, powerful concept for el-
egantly modelling many aspects of combinatorial optimisa-
tion problems. Also, there arenot too many different, and
evenstandard, ways of representing relations and relational
expressions. Therefore, I advocate that the compiler can
actually make a (systematic) empirical evaluation of candi-
date representations, using (real-life) training instances of the
problem. In the absence of such training instances, such a
compiler would simply be non-deterministic. Also, theoret-
ical studies such as[15] should be made for particular cases
of relations in order to obtain rules stating when a representa-
tion is advisable and when not, thereby reducing the volume
of such empirical studies by compilers.

4.2 Monogamy/Monoandry Version
Let us change the Stable Marriage problem to make it more
realistic. The numbers of women and men may now dif-
fer, so it may be impossible to arrange marriages for every-
one. However, let us keep the previous constraint that ev-
eryone may marry at most one person, of the opposite sex,
hence we call this themonogamy/monoandry version. The
sought marriages no longer form a bijection, but ratherpar-
tial functions between the sets of women and men. The
definition of marriage stability now has to be modified. A
monogamic/monoandric marriage is said to bestable if,
whenever spouses prefers some other partner, this partner
is married,and s/he prefers her/his spouse tos. To prevent
the problem from now having the empty set of marriages as a
solution, it has to become an optimisation problem. Various
‘happiness’ functions on the marriages can be defined, but I
here just take a simple one and maximise their number.

Figure 9 shows anESRA model of this new problem.6 A
cost function appeared, the lower bounds of the multiplicities
were changed from1 to 0, and the stability constraints were
rephrased to reflect the new definition. (The same stability
constraints could actually also have been used in the model
of Figure 8, because the new definition of stability implies
the original one in its context.) Model maintenance was thus
easy, because no representation issues had to be dealt with.

4.3 Polyandry Version
Imagine now a country where the law allows women to marry
up to 3 men, but men may marry only 1 woman. Also con-
sider that all women who signed up at the agency need to
marry (for some reason). We call this thepolyandry version.

6OPL doesnot have existential quantification.

enum Women, Men
nat RankW[Women,Men], RankM[Men,Women]
var Women 0:1 # 0:1 Men Marriage
maximise

count(w#m in Marriage)
subject to {

forall(w#m in Marriage) {
forall(o in Men) // stability 1

RankW[w,o] < RankW[w,m] =>
exists(p in Women: p#o in Marriage)

RankM[o,p] < RankM[o,w]
&
forall(p in Women) // stability 2

RankM[m,p] < RankM[m,w] =>
exists(o in Men: p#o in Marriage)

RankW[p,o] < RankW[p,m]
}

}

Figure 9: An ESRA model of the monogamic/monoandric
Stable Marriage problem

The sought marriages now form a fullrelation between the
sets of women and men. The definition of marriage stability
again has to be modified. A polyandric marriage is said to be
stable if, whenever spouses prefers some other partner, this
partneris married,and s/he prefersall her/his spouses tos. If
at least as many men as women have signed up at the agency,
the problem again becomes a decision problem.

Figure 10 shows anESRAmodel of this new problem.7 The
cost function disappeared, the multiplicities were changed,
and the second stability constraint was rephrased to reflect
the new definition, as the finalexists became aforall .
(The same stability constraints could actually also have been
used in the model of Figure 8, because the new definition
of stability implies the original one in its context.) Again, the
model maintenance was unburdened by representation issues.

5 Conclusion

Summary. I have argued that a high-level abstract con-
straint modelling language would enable an automatable em-
pirical determination — by a compiler — of how to ‘best’
represent the variables of a combinatorial optimisation prob-
lem. This compilation process would be based on (the few)
standard ways of (possibly redundantly) representing high-
level concepts such as relations, on theoretically or empiri-
cally gained knowledge about when to deploy which repre-
sentation, as well as on (real-life) training instances of the
problem. In the absence of training instances, such a com-
piler would simply be non-deterministic.

The advantage of such high-level modelling is that model
formulation and maintenance become much easier, because
unencumbered by early if not uninformed commitments to
representation choices. Automatable model reformulation
also becomes possible[8].

7OPL doesnot allow problem variables in the filtering formula
after the ‘such that’ (denoted ‘: ’) of universal quantifications.



enum Women, Men
nat RankW[Women,Men], RankM[Men,Women]
var Women 1:3 # 0:1 Men Marriage
solve {

forall(w#m in Marriage) {
// stability 1:
forall(o in Men) // stability 1

RankW[w,o] < RankW[w,m] =>
exists(p in Women: p#o in Marriage)

RankM[o,p] < RankM[o,w]
&
// stability 2:
forall(p in Women) // stability 2

RankM[m,p] < RankM[m,w] =>
forall(o in Men: p#o in Marriage)

RankW[p,o] < RankW[p,m]
}

}

Figure 10: AnESRAmodel of the polyandric Stable Marriage
problem

I have shown that a suitable first-order relational calculus
with sets is a good candidate for such a modelling language,
as it gives rise to very natural and easy-to-maintain modelsof
combinatorial optimisation problems.

Related Work. This research owes a lot to previous work
on relational modelling in formal methods and on ERA-style
semantic data modelling, especially to theALLOY object
modelling language[9], which itself gained much from theZ
specification notation[13] (and learned fromUML /OCL how
not to do it). Contrary to ERA modelling, I do not distinguish
between attributes and relations.

In constraint programming,OPL [14] stands out as a
medium-level constraint modelling language, andALMA
[2] is also becoming a very powerful notation, on top of
MODULA -2. Our ESRA shares with them the quest for a
practical declarative modelling language based on a strongly-
typed (full) first-order logic with arrays (and with the lookof
an imperative language), while dispensing with such hard-
to-properly-implement and rarely-necessary (for constraint
modelling) ‘luxuries’ as recursion and unbounded quantifica-
tion. As shown,ESRAeven goes beyond them, by advocating
an abstract view of sets and relations.

Future Work. I plan to evolve theESRA report[3] into a
complete (and more formal) description of the concepts, syn-
tax, and semantics of the fullESRA language, as well as to
implement its semantics with anESRA-to-OPL compiler. I
can then tackle the joint consideration of the modelling and
the solver parameterisation, as well as automated reformula-
tion at theESRA level. The language could be extended with
a syntax for declaratively expressing symmetries[6], so that
the compiler can generate suitable code for breaking them.
Also, a graphical language could be developed for the vari-
able modelling, including the multiplicity constraints onrela-
tions, so that only the cost function and the constraints would

need to be textually expressed.

